Extremwertaufgabe - ein komplexes Beispiel

Aus einem rechteckigen Stück Glas (gelbe Fläche) mit Länge k > 1
und Breite 3 ist ein Stück herausgebrochen. Die Bruchkante ist eine Parabel,
die für x = 0 eine waagerechte Tangente besitzt.

a) Wie lautet die Gleichung der Parabel

b) In welchen Fällen besitzt das blaue Rechteck maximalen Inhalt

```
> restart;
> with(plots):
> assume(k>1);
> p:=(x,k)->((k-1)/9)*x^2+1;
```

$$p := (x, k) \to \frac{1}{9}(k-1)x^2+1$$

```
> Glas:=[[0,0],[3,0],[3,2],[0,2]];
```

$$Glas := [[0, 0], [3, 0], [3, 2], [0, 2]]$$

```
> u:=p(2,2);
```

$$u := \frac{13}{9}$$

```
> Flaeche:=[[2,0],[3,0],[3,u],[2,u]];
```

$$Flaeche := \left[[2, 0], [3, 0], \left[3, \frac{13}{9}\right], \left[2, \frac{13}{9}\right]\right]$$

```
> pf:=plots[polygonplot](Flaeche,color=blue,style=patch):
> pg:=plots[polygonplot](Glas,color=yellow,style=patch):
> pp:=plot(p(x,2),x=0..3,y=0..3,color=red,thickness=3):
> display({pg,pf,pp},axes=framed);
```



oben : Der Plot zum Problem
unten : die Flächenfunktion

```
> z:=(x,k)->(3-x)*(((k-1)/9)*x^2+1);
```

$$z := (x, k) \to (3 - x)\left(\frac{1}{9}(k - 1)\,x^2 + 1\right)$$

```
> z1:=unapply(simplify(diff(z(x,k),x)),(x,k));
```

$$z1 := (x, k\sim) \to -\frac{1}{3}x^2\,k\sim + \frac{1}{3}x^2 - 1 + \frac{2}{3}k\sim\,x - \frac{2}{3}x$$

```
> z2:=unapply(simplify(diff(z1(x,k),x)),(x,k));
```

$$z2 := (x, k\sim) \to -\frac{2}{3}k\sim\,x + \frac{2}{3}x + \frac{2}{3}k\sim - \frac{2}{3}$$

```
> solve(z1(x,k)=0,x);
```

$$\frac{1}{2}\,\frac{2\,k\sim - 2 + 2\sqrt{k\sim^2 - 5\,k\sim + 4}}{k\sim - 1},\ \frac{1}{2}\,\frac{2\,k\sim - 2 - 2\sqrt{k\sim^2 - 5\,k\sim + 4}}{k\sim - 1}$$

```
> l1:=1/2*(-2+2*k+2*sqrt(4-5*k+k^2))/(k-1);
```

$$l1 := \frac{1}{2}\,\frac{2\,k\sim - 2 + 2\sqrt{k\sim^2 - 5\,k\sim + 4}}{k\sim - 1}$$

```
> loesung1:=simplify(l1);
```

$$loesung1 := \frac{k\sim - 1 + \sqrt{k\sim^2 - 5\,k\sim + 4}}{k\sim - 1}$$

```
> l2:=1/2*(-2+2*k-2*sqrt(4-5*k+k^2))/(k-1);
```

$$l2 := \frac{1}{2}\,\frac{2\,k\sim - 2 - 2\sqrt{k\sim^2 - 5\,k\sim + 4}}{k\sim - 1}$$

```
> loesung2:=simplify(l2);
```

$$loesung2 := \frac{k\sim - 1 - \sqrt{k\sim^2 - 5\,k\sim + 4}}{k\sim - 1}$$

```
> Loes1:=unapply(loesung1,k);
```

$$Loes1 := k\sim \to \frac{k\sim - 1 + \sqrt{k\sim^2 - 5\,k\sim + 4}}{k\sim - 1}$$

> **Loes2:=unapply(loesung2,k);**

$$Loes2 := k\sim \rightarrow \frac{k\sim - 1 - \sqrt{k\sim^2 - 5\,k\sim + 4}}{k\sim - 1}$$

Fall 1 : k = 3 , kein Kandidat für relative Extremwerte, warum ?

> **x1:=Loes1(3);**

$$x1 := 1 + \frac{1}{2}\sqrt{-2}$$

> **x2:=Loes2(3);**

$$x2 := 1 - \frac{1}{2}\sqrt{-2}$$

daher muß das Maximum am Rand liegen :

> **'z(0,3)'=z(0,3);**

$$z(0, 3) = 3$$

> **'z(3,3)'=z(3,3);**

$$z(3, 3) = 0$$

also liegt das Maximum bei x = 0

Fall 2 : k = 4, nur ein Kandidat, denn :

> **x1:=Loes1(4);**

$$x1 := 1$$

> **x2:=Loes2(4);**

$$x2 := 1$$

> **'z(0,4)'=z(0,4);**

$$z(0, 4) = 3$$

> **'z(3,4)'=z(3,4);**

$$z(3, 4) = 0$$

> **'z(1,4)'=z(1,4);**

$$z(1, 4) = \frac{8}{3}$$

also liegt das Maximum wieder am Rand, bei x = 0
(Warum ist x = 1 keine Lösung ?)

Fall 3 : k = 5

```
> x1:=simplify(Loes1(5));
```

$$x1 := \frac{3}{2}$$

```
> x2:=simplify(Loes2(5));
```

$$x2 := \frac{1}{2}$$

```
> 'z2(x1,5)'=z2(x1,5);
```

$$z2(x1, 5) = \frac{-4}{3}$$

```
> 'z(x1,5)'=z(x1,5);
```

$$z(x1, 5) = 3$$

also liegt bei x1 ein relatives Maximum vor

```
> 'z(0,5)'=z(0,5);
```

$$z(0, 5) = 3$$

also sind x = 0 (Randwert) und x1 (lokal) globale Maxima und
damit Lösungen des Problems

```
> 'z2(x2,5)'=z2(x2,5);
```

$$z2(x2, 5) = \frac{4}{3}$$

```
> 'z(x2,5)'=z(x2,5);
```

$$z(x2, 5) = \frac{25}{9}$$

also ist x2 relatives Minimum.

Fall 4 : k = 10

```
> x1:=simplify(Loes1(10));
```

$$x1 := 1 + \frac{1}{3}\sqrt{6}$$

```
> x2:=simplify(Loes2(10));
```

$$x2 := 1 - \frac{1}{3}\sqrt{6}$$

```
> 'z2(x1,10)'=z2(x1,10);
```

$$z2(x1, 10) = -2\sqrt{6}$$

```
> 'z(x1,10)'=simplify(z(x1,10));
```

$$z(x1, 10) = -\frac{2}{9}(-6 + \sqrt{6})(4 + \sqrt{6})$$

```
> xmax:=evalf(x1);
```

$$xmax := 1.816496581$$

```
> ymax:=evalf(simplify(z(x1,10)));
```

$$ymax := 5.088662106$$

```
> 'z2(x2,10)'=z2(x2,10);
```

$$z2(x2, 10) = 2\sqrt{6}$$

```
> 'z(x2,10)'=simplify(z(x2,10));
```

$$z(x2, 10) = -\frac{2}{9}(6 + \sqrt{6})(-4 + \sqrt{6})$$

```
> xmin:=evalf(x2);
```

$$xmin := .1835034191$$

```
> ymin:=evalf(simplify(z(x2,10)));
```

$$ymin := 2.911337891$$

```
> 'z(0,10)'=z(0,10);
```

$$z(0, 10) = 3$$

```
> 'z(3,10)'=z(3,10);
```

$$z(3, 10) = 0$$

also liegt im Fall k = 10 an der Stelle xmax das relative und gleichzeitig globale Maximum

```
> pp3:=plot(z(x,3),x=0..3,y=0..6,color=blue,thickness=2):
> pp4:=plot(z(x,4),x=0..3,y=0..6,color=blue,thickness=2):
> pp5:=plot(z(x,5),x=0..3,y=0..6,color=blue,thickness=2):
```

```
> pp10:=plot(z(x,10),x=0..3,y=0..6,color=blue,thickness=2):
> display({pp3,pp4,pp5,pp10},axes=framed);
```



oben : die Funktionenschar für die betrachteten Fälle

Jetzt alle Fälle mit Rechteck :

1. k = 3

```
> o:=p(0,3);
```

$o := 1$

```
> Flmax:=[[0,0],[3,0],[3,o],[0,o],[0,0]];
```

$Flmax := [[0, 0], [3, 0], [3, 1], [0, 1], [0, 0]]$

```
> Glas:=[[0,0],[3,0],[3,3],[0,3]];
```

$Glas := [[0, 0], [3, 0], [3, 3], [0, 3]]$

```
> pfmax:=plots[polygonplot](Flmax,color=blue,style=patch):
> pg:=plots[polygonplot](Glas,color=yellow):
> p3:=plot(z(x,3),x=0..3,y=0..6,color=red,thickness=2):
> pp:=plot(p(x,3),x=0..3,y=0..6,color=black,thickness=2):
> display({pfmax,pg,p3,pp},axes=framed);
```



2. k = 4

```
> o:=p(0,4);
```

$o := 1$

```
> Flmax:=[[0,0],[3,0],[3,o],[0,o],[0,0]];
```

$Flmax := [[0, 0], [3, 0], [3, 1], [0, 1], [0, 0]]$

```
> Glas:=[[0,0],[3,0],[3,4],[0,4]];
```

$Glas := [[0, 0], [3, 0], [3, 4], [0, 4]]$

```
> pfmax:=plots[polygonplot](Flmax,color=blue,style=patch):
> pg:=plots[polygonplot](Glas,color=yellow):
> p3:=plot(z(x,4),x=0..3,y=0..6,color=red,thickness=2):
> pp:=plot(p(x,4),x=0..3,y=0..6,color=black,thickness=2):
> display({pfmax,pg,p3,pp},axes=framed);
```



3. k = 5

```
> o:=p(3/2,5);
```

$o := 2$

```
> Flmax:=[[3/2,0],[3,0],[3,o],[3/2,o],[3/2,0]];
```

$$Flmax := \left[\left[\frac{3}{2}, 0\right], [3, 0], [3, 2], \left[\frac{3}{2}, 2\right], \left[\frac{3}{2}, 0\right]\right]$$

```
> o:=p(0,5);
```

$o := 1$

```
> Flmax1:=[[0,0],[3,0],[3,o],[0,o],[0,0]];
>
```

$$Flmax1 := [[0, 0], [3, 0], [3, 1], [0, 1], [0, 0]]$$

```
> Glas:=[[0,0],[3,0],[3,5],[0,5]];
```

$$Glas := [[0, 0], [3, 0], [3, 5], [0, 5]]$$

```
> pg:=plots[polygonplot](Glas,color=yellow,style=patch):
> pfmax:=plots[polygonplot](Flmax,color=blue,style=patch):
> p3:=plot(z(x,5),x=0..3,y=0..6,color=red,thickness=2):
> pp:=plot(p(x,5),x=0..3,y=0..6,color=black,thickness=2):
> pfmax1:=plots[polygonplot](Flmax1,color=red,style=line,thickness=3):
> display({pg,pfmax1,pfmax,p3,pp},axes=framed);
```



4. k = 10

```
> o:=p(1+1/3*sqrt(6),10);
```

$$o := \left(1 + \frac{1}{3}\sqrt{6}\right)^2 + 1$$

```
> Flmax:=[[1+1/3*sqrt(6),0],[3,0],[3,o],[1+1/3*sqrt(6),o]];
```

$$Flmax := \left[\left[1+\frac{1}{3}\sqrt{6}, 0\right], [3, 0], \left[3, \left(1+\frac{1}{3}\sqrt{6}\right)^2+1\right], \left[1+\frac{1}{3}\sqrt{6}, \left(1+\frac{1}{3}\sqrt{6}\right)^2+1\right]\right]$$

```
> Glas:=[[0,0],[3,0],[3,10],[0,10]];
```

$$Glas := [[0, 0], [3, 0], [3, 10], [0, 10]]$$

```
> pfmax:=plots[polygonplot](Flmax,color=blue,style=patch):
> pg:=plots[polygonplot](Glas,color=yellow):
> p3:=plot(z(x,10),x=0..3,y=0..10,color=red,thickness=2):
> pp:=plot(p(x,10),x=0..3,y=0..10,color=black,thickness=2):
> display({pfmax,pg,p3,pp},axes=framed);
```